



Let's Code Blacksburg's Arduino Cookbook



Version 2018-06-26

By Monta Elkins, Eddie Sheffield and Thomas Weeks

Let's Code Blacksburg 2018  (AS)

Online PDF: <https://github.com/LetsCodeBlacksburg/arduino-recipes/>

How To Cook (use this Arduino cookbook)

What Is This Cookbook?

This Arduino circuits and programming instruction guide is organized into a “cookbook” style layout. The cookbook illustrates how to create and write various arduino based circuits and programs. These instructions are organized into “Recipes” or instruction guides that can be combined in different ways to come up with new creations. All the code from this cookbook is in our code repository at <https://github.com/LetsCodeBlacksburg/arduino-recipes/>



How This All Works:

For example, want to make a robot? Learn how to move a servo motor with our *Servo recipe*. Then learn how to read an *Ultrasonic Eyes recipe*. Combine them to make a collision avoiding robot! Or use the *Potentiometer (knob) recipe* plus the *Sound/speaker recipe* and you've got a musical instrument! The only limit is your own imagination!

The circuit and programming recipes, just like this intro, each feature a **What**, **How** and **Fail** section – quickly telling you what you're doing, illustrating how to do it, and what to look for if something fails. Identifying failure is important as it is what enables you to “fail fast”, learn and move on quickly to success. Failure is good! Without it, learning is difficult and much less satisfying.

Failure:

A bit more on failure – one needs to understand and embrace that ***Failure is a natural part of learning***, but before you can learn and move on from failure, you must:

- 1) **Recognize** failure has or is occurring
- 2) **Step Back** and determine (or guess at) the nature or root of the problem
- 3) **Test Your Assumptions** (of the problem) and correct assumptions when needed
- 3) **Identify & Address the Problem**, work around it or create a new way of accomplishing the desired outcome.

Also understand that the failure sections are *not* definitive documents on every possible failure you can encounter for a given recipe. No “failure guide” can easily contain every possible failure for a given circuit or technical process. The failure section is more a guide to help get your thinking cap back on straight and think about the nature of the problem, and what quick fixes or work-arounds might get you back on track!

Don't be afraid of failure! Identify it, embrace it, learn from it and move on.

“Negative results are just what I want. They’re just as valuable to me as positive results. I can never find the thing that does the job best until I find the ones that don’t.”

— Thomas Edison

Or as one of our class instructors cries out to his students,

“Fail fast, fail cheap!”

— Monta Elkins



Get Cooking! With Let's Code Blacksburg Cookbook Recipes:

Need Some Ideas To Get Cooking? Check Out These Recipes !!!

This whole cookbook is designed like legos. You put together the various recipes to create thousands of cool, unusual inventions! For example, you can mix and match the recipes like this :



- Use the *Light Sensor recipe* to detect afternoon sunlight and close the blinds with a servo!
- The *People Motion Sensor recipe* you can detect movement, sound and alarm and then use the *Ethernet shield recipe* to send a TXT message or email!
- With just the *Light Sensor recipe* and a speaker you can make eerie sci-fi music!
- Combine multiple servos and potentiometers recipes to control a two axis robotic, popsicle stick arm (Advanced project. Ask for a special recipe handout if interested.)
- Use the *Wired or Wireless Web recipe* to share real time measurements to the Web or send via an SMS text message or email!

The only limits are your own imagination, available memory, and I/O pins! :)

* - Recommended or required for beginners

** - Really fun *and* easy for new beginners

Pg	Recipe Name	Description
4)	Arduino Hardware & Pinout Overview	See where to hook things up.
5)	Arduino Cheat Sheet	All the commands you need to at least know about.
6)	*Installing Arduino Software & Drivers Recipe	Required on new PCs to talk to the Arduino.
7)	*TEST: LED (light) Blink Recipe	Test compiles and uploads code to blink a built in LED.
9)	OUTPUT: Serial Monitor Recipe	Outputs text back to the PC. Great for live troubleshooting.
11)	*BUILD: Breadboard or Protoboard Recipe	Breadboards are your pallet for creating temporary circuits.
13)	*INPUT: Push Button	the most simple way of taking input.
15)	*INPUT: Potentiometer Recipe	A knob or "pot", a adjustable resistor for creating a variable voltage.
16)	**BUILD: LED Chase Light Recipe	Wire up and blink a chase light circuit.
18)	CONTROL: Servo Recipe	A servo is a PWM, digital motor that you can control the angular position of.
20)	CONTROL: Robot DC Motor Control	Using the KeyesL298 H-Bridge motor controller w/DC motors.
24)	INPUT: Ultrasonic "Eyes" Range Sensor	Ultrasonic sensor to measure distance to objects up to 24" out.
26)	INPUT: Line Following IR Sensor	Using a three-element infrared line sensor.
28)	CONTROL: Potentiometer Control of a Servo's Position Recipe	read a pot, move a servo.
30)	*OUTPUT: Buzzer Alarms	Using piezo buzzers for beeps, buzzing and tones.
32)	**OUTPUT: Sound Generation with Arduino	More advanced tones, sounds and music with speaker.
36)	OUTPUT: Playing MP3 Audio	Playing sounds/music w/the DFPlayer mini MP3 player module.
37)	**INPUT: Light Sensor	Detecting light intensity with a CdS photo-sensor.
39)	INPUT: DHT-22 Relative Humidity/Temperature Sensor	Detecting humidity & temperature via DHT sensor.
41)	**INPUT: People Motion Sensor	PIR motion sensor (senses motion up to 20ft away).
43)	OUTPUT: 7 Segment LED Display w/TM1637 Module	Displaying "digital clock" numbers with LEDs.
44)	INPUT / OUTPUT: The LCD Display / Keypad shield	Displaying text and taking keypress inputs.
51)	COMM: Wired Web & Email Communication	Using Wired Ethernet Networking for Web & Email.
59)	COMM: ESP-01 Wireless Web & Email Communications	Using WiFi networking for Web & Email.

Bold Italics = New


The Authors:

Eddie Sheffield, Thomas “Tweeks” Weeks, and Monta Elkins*



* - the guy who came up with the idea of the arduino cookbook concept... convinced us all it would work.. and then got his friends to do all the heavy lifting. Thanks Monta! ;)

The Arduino Microcontroller



Arduino Programming Cheat Sheet

Primary source: Arduino Language Reference
<http://arduino.cc/en/Reference/>

Structure & Flow

```
Basic Program Structure
void setup() {
  // Runs once when sketch starts
}

void loop() {
  // Runs repeatedly
}

Control Structures
if (x < 5) { ... } else { ... }
while (x < 5) { ... }
for (int i = 0; i < 10; i++) { ... }
break; // Exit a loop immediately
continue; // Go to next iteration
switch (var) {
  case 1:
    ...
  case 2:
    ...
  default:
    ...
}

return x; // x must match return type
return; // For void return type

Function Definitions
<ret. type> <name>(<params>) { ... }
e.g. int double(int x) {return x*2;}
```

Variables, Arrays, and Data

```
Data Types
boolean true | false
char -128 - 127, 'a' '$' etc.
unsigned char 0 - 255
byte 0 - 255
int -32768 - 32767
unsigned int 0 - 65535
word 0 - 65535
long -2147483648 - 2147483647
unsigned long 0 - 4294967295
float -3.4028e+38 - 3.4028e+38
double currently same as float
void i.e., no return value

Strings
char str1[8] =
  {'A', 'r', 'd', 'u', 'i', 'n', 'o', '\0'};
// Includes \0 null termination
char str2[8] =
  {'A', 'r', 'd', 'u', 'i', 'n', 'o'};
// Compiler adds null termination
char str3[] = "Arduino";
char str4[8] = "Arduino";
```

Operators

```
General Operators
= assignment
+ add - subtract
* multiply / divide
% modulo
== equal to != not equal to
< less than > greater than
<= less than or equal to
>= greater than or equal to
&& and || or
! not

Compound Operators
++ increment
-- decrement
+= compound addition
-= compound subtraction
*= compound multiplication
/= compound division
&= compound bitwise and
|= compound bitwise or

Bitwise Operators
& bitwise and | bitwise or
^ bitwise xor ~ bitwise not
<< shift left >> shift right

Pointer Access
* reference: get a pointer
* dereference: follow a pointer
```

Built-in Functions

```
Pin Input/Output
Digital I/O - pins 0-13 A0-A5
pinMode(pin, mode)
  [INPUT, OUTPUT, INPUT_PULLUP]
int digitalWrite(pin, value)
digitalWrite(pin, [HIGH, LOW])

Analog In - pins A0-A5
int analogRead(pin)
analogReference([DEFAULT, INTERNAL, EXTERNAL])

PWM Out - pins 3 5 6 9 10 11
analogWrite(pin, value)

Advanced I/O
tone(pin, freq_Hz)
noTone(pin)
shiftOut(dataPin, clockPin,
  [MSBFIRST, LSBFIRST], value)
unsigned long pulseIn(pin,
  [HIGH, LOW])

Type Conversions
char(val) byte(val)
int(val) word(val)
long(val) float(val)

External Interrupts
attachInterrupt(interrupt, func,
  [LOW, CHANGE, RISING, FALLING])
detachInterrupt(interrupt)
interrupts()
noInterrupts()

Time
unsigned long millis()
  // Overflows at 50 days
unsigned long micros()
  // Overflows at 70 minutes
delay(msec)
delayMicroseconds(usc)
```

Libraries

```
Serial - comm. with PC or via RX/TX
begin(long speed) // Up to 115200
end()
int available() // #bytes available
int read() // -1 if none available
int peek() // Read w/o removing
flush()
print(data) println(data)
write(byte) write(char * string)
write(byte * data, size)
SerialEvent() // Called if data rdy

SoftwareSerial.h - comm. on any pin
SoftwareSerial(rxPin, txPin)
begin(long speed) // Up to 115200
listen() // Only 1 can listen
isListening() // at a time.
read, peek, print, println, write
// Equivalent to Serial library

EEPROM.h - access non-volatile memory
byte read(addr)
write(addr, byte)
EEPROM[index] // Access as array

Servo.h - control servo motors
attach(pin, [min_us, max_us])
write(angle) // 0 to 180
writeMicroseconds(us)
  // 1000-2000; 1500 is midpoint
int read() // 0 to 180
bool attached()
detach()


Wire.h - I2C communication
begin() // Join a master
begin(addr) // Join a slave @ addr
requestFrom(addr, count)
beginTransmission(addr) // Step 1
send(byte) // Step 2
send(char * string)
send(byte * data, size)
endTransmission() // Step 3
int available() // #bytes available
byte receive() // Get next byte
onReceive(handler)
onRequest(handler)
```



by Mark Liffiton

Adapted from:

- Original: Gavin Smith
- SVG version: Frederic Dufourg
- Arduino board drawing: Fritzing.org



Installing Arduino Software & Drivers Recipe needed on a fresh PC to talk to the Arduino

What:

This is the process that installs two things; the Arduino IDE or programming software that you use to program and upload code to the Arduino, and the USB/serial port drivers (if needed on Mac and Windows) to “talk” to the board over the USB interface.

NOTE: If you can not get this working (test it using the Blink Recipe), then you will not be able to work with any other recipes in this cookbook.

How:

For Linux

•RedHat:

```
# yum -y install arduino #(reqs: uisp avr-libc avr-gcc-c++ rxtx avrdude)
```

•Ubuntu:

```
$ sudo apt-get install arduino #(reqs uisp avr-libc gcc-avr avrdude librx-tx-java)
```

•or for other installs or source based installs, go here:

<http://playground.arduino.cc/Learning/Linux>

For Mac/OSX:

•Download & Install Software from: <http://arduino.cc/en/Guide/MacOSX#.UwGmXXWqYY0>

For Windows:

•Download & install the software from: <http://arduino.cc/en/Guide/Windows#.UwGlyHWqYY0>

Fail:

Linux T-Shooting:

•check permissions of /dev/ttyUSB0 or /dev/ttyACM0 (user needs r/w or 777 access)

•May have to open port permissions to:

```
# usermod -a -G uucp,dialout,lock $USER
```

or may have to tempfix as root :

```
# chmod 777 /dev/ttyUSB0
```

Windows T-Shooting:

•Make sure the special USB serial port drivers are installed

<http://arduino.cc/en/Guide/UnoDriversWindowsXP#.UwGbtHWqYY0>

•Check/fix COM port settings

Mac T-Shooting:

•Make sure the special USB serial port drivers are installed

<http://arduino.cc/en/Guide/MacOSX#toc3>

•check device permissions (similar to Linux)


TEST: LED (light) Blink Recipe test compiles and uploads code to blink a built in LED

What:

This process simply compiles and uploads code to the Arduino for execution and blinks a light when it succeeds. It's the easiest and fastest (fail fast) test method to verify you can talk to your arduino.

There is a small LED (the light) connected to pin 13 of the Arduino. When that pin is 'high' (meaning +5 volts for this Arduino clone), the LED lights.

NOTE: Arduino Software and driver should be installed. (see Installing Arduino Software and Drivers recipe).




Arduino Clone with power and "blink" light

How:

Select appropriate port
“Tools / Serial port” (see Fail section if the Serial port menu is ghosted).

Select Arduino Uno.
“Tools / Board / Arduino Uno”.

Load blink onto the arduino, “File / Examples / Basics / Blink” (see right).


Click on the  upload icon to compile and upload your program.

Several lights will blink during the upload, then your program runs.

Look for the steady light blinking here.

Change both lines that say **delay(1000);** to **delay(100);** and reupload the program.

Look for the light to blink faster. This shows that the program changes you made are actually uploaded to the Arduino.



Loading the "blink" program in the Arduino IDE

This blinking is done by the command **digitalWrite(led, HIGH);** which sends a “HIGH” 5volts to the pin# in the variable “led”. When pin 13 is 'LOW' (meaning connected to ground) the LED (light) is off. This is done by the command **digitalWrite(led, LOW);**

NOTE: Anything following “//” on a line is considered comments and ignored.

Fail:

- If you can not write to the arduino, get some error, or the serial port is ghosted:
 - Verify the correct serial port
 - Unplug Arduino and list serial ports, then plug up Arduino and list serial ports again.
 - An additional serial port should appear.
 - That new serial port should be the Arduino port.OR
 - Try plugging Arduino into a different port USB port
- OR

- See the fail section of the “Installing Arduino Software & Drivers” recipe (port/permissions/drivers)
- OR

- Test with a different cable and Arduino board.
- OR.. and only as a last resort in Linux...

- Try running the arduino program as root (gets around all permission errors. Usually for testing only)

OUTPUT: Serial Monitor Recipe outputs text back to the PC. Great for live troubleshooting

What:


The serial output on the arduino can be used to echo or print real time program data back to the PC over the serial USB port. This is very handy for troubleshooting or looking at run time values, states and problem code.

How:

Upload and execute the following sketch on the Arduino.

```
void setup() {  
  // initialize serial communications at 9600 bps:  
  Serial.begin(9600);  
}  
  
void loop() {  
  
  Serial.println("Hello World");  
  
  // wait 1 second before the next loop  
  delay(1000);  
}
```

Then open the Serial Monitor by selection “Tools / Serial Monitor” or clicking its icon on the far right .  After which you should see the line “Hello World” appear repeatedly in the serial monitor window.



Serial.begin(9600); opens the serial port at on the Arduino and sets its speed to 9600 baud
The **Serial.println()** statement prints a line followed by a newline. When the Serial Monitor is opened it watched for characters appearing on the laptop's serial (or USB serial) port and prints them in the serial monitor window.

Fail:

- Be sure that you have the **Serial.begin(9600);** defined in the **void setup () {** code block.
- Look for RX and TX lights to blink on Arduino rapidly during upload.
- If program is running successfully look for the TX light to blink once per second, showing that it is “trying” to transmitting data back to the laptop.
- Double check your Serial Monitor settings for both port and speed.

NOTE: you have to restart the Serial Monitor after each upload of a new program, because the upload process uses the same serial port connection.

Breadboard or Protoboard Recipe breadboards are your pallet for creating temporary circuits


What:

How does a breadboard work?

Wire pins, pushed into the breadboard are connected together as shown in the schematic below. This allows the quick building and testing of circuits.

NOTE: Unplug the Arduino from the laptop (and any other power supply while making and verifying connections).


Connection wire colors do not matter; but traditionally power (+) wires are red and ground (-) wires are blue (or black). The (+) red breadboard row and (-) blue breadboard row gives you a common hookup location on the breadboard for power (red) and ground (blue, GND). Using them also makes troubleshooting a little easier.



Average "breadboard" or "protoboard" (w/ horizontal +/- power rails) which connects your circuit's 5V and GND with the Arduino'.

Under the white plastic of the breadboard you see that the holes are connected. This is what makes working with a breadboard like legos for electronics.

WARNING: Never connect or short +5v power or (+) to GND or (-). This short will probably blow your laptop's USB port and damage other hardware. Not to mention upsetting your instructor and being mocked by your classmates




Protoboard schematic, showing the long +/5V and -/GND power rails, and the short "rows" where you push in your components. -Wikipedia


How:

To build a circuit, you normally:


- 1) start with a schematic diagram (left) which illustrates what is connected to what,
- 2) orient your components correctly (note the polarity (+ and -) on the LED light)
- 3) and connect them together on the breadboard:



Schematic diagram




LED polarity



Blank breadboard

This is what the assembled circuit might look like:



-Batt to ground/-rail to -LED, +LED to Resistor to +rail to +Batt

Fail:

NEVER connect + to - directly. This will damage the breadboard, the wires, the battery and possibly damage your Arduino and even your USB port or PC!

INPUT: Push Button

the most simple way of taking input

What:

Other than using a raw wire, the momentary push button is the most simple form of input there is. It can be wired “HIGH” (to result in a logical 1, or True) or wired “LOW” (to result in a logical 0, or false) condition when using the `digitalRead(pin)` command.


How:

Most people think you can just hook it up to an input and wire it to Vcc (5v) or ground and it just works. Due to digital logic “floating inputs”, it's not quite that simple, and as such you usually want at least a single 10k resistor to pull the un-pushed state of the button HIGH or LOW.


If you want a logical “1” or True state when you push the button, then you should tie the un-pushed side of the switch to GND (through a resistor) so that the un-pushed logic level floats low. If you do not do this, then it can float high or low and create problems for your program.

Always disconnect the Arduino from the USB & power before connecting anything new. Wire it as seen here so that un-pushed, pin 3 “sees” the logic LOW (or GND) when the button is left unpressed, but as soon as the button is pressed, the pin 3 sees the Vcc (5V) or logic HIGH.

WARNING: Be very careful that you do not wire the switch directly across the 5V and GND or skip using the resistor. This will short the Arduino's 5V to GND which will make the switch very hot, possibly smoke/melt the breadboard, and damage the Arduino. If unsure of your connections, always get a TA- helper to look over your work!



Mechanical and electrical diagrams for a push button switch



Here's the code to make it work. We're just blinking LED 13 using an if() conditional to show the state of the button.


```
const int buttonPin = 3;
const int ledPin = 13;

void setup() {
  // initialize serial communication at 9600 bits per second:
  Serial.begin(9600);
  pinMode(buttonPin, INPUT);
  pinMode(ledPin, OUTPUT);
}

// the loop routine runs over and over again forever:
void loop() {

  if (digitalRead(buttonPin)) {           // If button push is true
    digitalWrite(ledPin, HIGH);           // light LED
    Serial.println("Button Pushed!");     // and print on serial port
    delay(10);
  }
  else {
    digitalWrite(ledPin, LOW);             // otherwise, turn off LED
    delay(1);
  }

  delay(10);                             // delay in between reads for stability
}
```

Once you compile and upload the code, be sure to press the serial monitor icon  to pull up the real time serial data coming back from the Arduino so you can see the report of the push button state. However, the LED should also light, indicating the input state change.

Fail:

Nothing happens when you press the button:

- If the button seems to do nothing, make sure you have it oriented correctly. If you wire it 90 degrees off, it will act like the button is constantly pushed and the state will not change.
- You have wired the resistor, 5V or GND wrong.
- Your wiring and buttonPin setting do not match.
- Bad coding.


The button or circuit gets hot or smokes:

- STOP! Unplug everything and have an instructor come over and look at your wiring. You have probably shorted 5V and GND. Very bad juju.

INPUT: Potentiometer Recipe a knob or “pot” is a adjustable resistor for creating a variable voltage

What:

The potentiometer is just a variable resistor. In most implementations here you will see it used as a knob that will give you a variable voltage (e.g. 0 – 5 volts) to control things hooked to the arduino. The outer two pins, left(1) and right(3), get hooked to ground (GND) and +5 volts, and in this configuration the middle pin (2) will provide a 0 - 5 volt range that can in turn be applied to an arduino analog input such as A0. If operated like this, the arduino will read that value in (when instructed) and convert any analog voltage (at that moment) to a number between 0 - 1023 through a process called analog to digital conversion, much like an MP3 recorder does for audio.




Mechanical and schematic diagrams for a potentiometer

How:

Firmly insert the potentiometer (also called a “pot”) into the breadboard. Connect the leftmost side to ground (GND or -) and the rightmost side to power (+5v or +, or +3.3v if configured for 3.3v operation). The middle connector is the output of the pot in this case. Connect the middle wiper arm to an analog in pin on the Arduino. The A0 input is good.

Verify the circuit by reading and printing the pot value.

```
void setup() {  
  // initialize serial link at 9600 bps:  
  Serial.begin(9600);  
}  
  
void loop() {  
  int sensorValue0;  
  sensorValue0 = analogRead(A0); // read the pot input  
  Serial.print ("Pot 0 value="); // print it  
  Serial.println(sensorValue0);  
  
  // wait 1 second before the next loop  
  delay(1000);  
}
```



Turn the pot left and right, the printed value should go from (near) 0 to (near) 1023

Fail:

- Verify the pot is seated firmly in the protoboard
- Verify 5 volts across the pot's outer pins with a multi-meter
- Are you reading the input value into a variable?
- Are you printing the correct variable for testing?
Discuss 'deadband' and “print on change”

BUILD: LED Chase Light Recipe

wire up and blink a chase light circuit

What:

The goal of this recipe is to learn how to hook up multiple LEDs and resistors to the arduino in order that make them strobe back and forth to form a chase light. Add in an optional potentiometer and you can control things like LED chase speed or LED brightness.


You will need:

- a breadboard
- 6-7 LEDs
- a 330 ohm resistor
- 5 or 10k potentiometer (optional)

How:

Before hooking anything up, first note that the LED has a longer leg (positive) and a short leg (negative). Unlike a light bulb, LED lights have + and – polarity and need to be correctly connected.

WARNING: If you get the polarity of an LED hooked up backwards it simply won't light. However, get the polarity right but *without a current limiting resistor* and you can blow it. Please don't blow our LEDs or we'll call you “smokey”. :^)



Connect Single LED + Resistor:

First, start off by hooking up just one LED and current limiting resistor and get that working on the arduino's digital output pin 12. To do this:

- Connect the GND on the Arduino to the 330 ohm resistor and then the resistor down to the breadboard's - or blue row. This is blue row is where you will connect the - side of the LED.


NOTE: Leave +5V disconnected from the breadboard for this recipe. We're using the + (red) row for something else in this circuit.


- Connect LED's positive (long) leg to digital pin 12
- Connect LED's negative (short) leg to - blue row of the breadboard
- Connect other side of the resistor to GND (blue row) on breadboard
- Load the “blink” program and change `digitalWrite(13);` to pin "12"
- Upload & run

Chase Lights:

After you have one LED up and running, hook up the remaining 4-5 LEDs the same way to pins 8, 9, 10 and 11, all back through the same resistor through the blue(-) power strip. See photo (right).

Hook up:

- LED positive (long) legs to digital pins 8, 9, 10, 11, 12 (top “DIGITAL” input/output section)
- LED negative (short) legs to the resistor's common blue row
- Modify program **void setup()** section to configure LED pins 8-12 as outputs
- Compile and upload 



TEST: Test to see if you can light up each LED with the digitalWrite() command

Use Potentiometer (knob) Value For Timing:

- Add potentiometer to breadboard to control variable chase light speed
- Wire pot leg pin 1 (left pin) to ground (GND or 0v) on arduino
- Wire pot leg pin 2 (middle) to A0 or "Analog0" (on “ANALOG IN” header (bottom right))
- Wire pot leg pin 3 (right pin) to +5v on arduino
- Use "analogRead(0);" function read or *sample* the pot value (0-5V maps to value 0 -1023)
- Replace blink's
delay(1000);

in milli-seconds with:

delay(analogRead(0));

to use the pot read 0-1023 value as the new delay value between LED flashes.

- Compile, upload and run code
- Twist knob to adjust chase light speed (delay)

Try This: You can either read the analogRead(0) just once at the beginning of your LED flashes (using it for each LED on/off cycle), or re-read the pot for each LED flash cycle. Try it both ways. Observe the difference.

Try This: See data from the arduino on your PC in real time with **Serial.print(analogRead(0));** to see your pot value in the serial console

Fail:

- LED is not lighting: Check the polarity or try new LED (another “smokey” may have used it ;)
- Pot does nothing: Make sure you have the pot's pins hooked up to GND and +5v correctly or that you're reading the correct analog input. If problems, then Serial.println() the value of the pot to verify it's working as expected.

CONTROL: Servo Recipe a servo is a digitally driven (PMW) motor that you can control the angul of.

What:

Connect and control a servo motor (a digitally, position controlled motor) with the Arduino with a potentiometer (knob). You will read the value of the pot, and based on that value, change the position of the servo motor using one of the arduino's PWM (pulse width modulation) outputs.

How:

Use a 3 pin header to connect the servo to the protoboard.

- Connect the brown wire on the servo to ground.
- Connect the red wire on the servo to the Vin pin on the Arduino.
- Connect the orange wire to pin 9 on the Arduino.

WARNING: Placing a 330 ohm buffering resistor on the servo input line (orange) is a good idea to help protect the circuit in case of mis-wiring. “Only you can prevent arduino fires Smokey.”



Use the following code to test the servo.

```
#include <Servo.h>
// create a servo object
Servo servo0;


void setup() {
  servo0.attach(9); // servo is attached to pin 9
}

void loop() {

  servo0.write(60); // tell servo to go to the 60 degree position
  delay(1000);      // wait 1 second
  servo0.write(120); // tell servo to go to the 120 degree position
  delay(1000);      // wait 1 second

}
```

Look for the preceding code to move the servo to the “60 degree” position, wait 1 second, then move the servo to the “120 degree” position. The actual movement degree may vary somewhat depending on the servo. The initial position of the plastic servo arm that presses onto the servo toothed shaft may be changed by gently pulling it up, off the servo, turning it and then pressing it back down, re-engaging the “teeth” in a different rotational position.



How:

The position of a servo is set by sending it a 1 – 2 millisecond pulse. A 1ms pulse represents approximately 0 degrees of servo rotation. A 1.5 ms pulse represents approximately 90 degrees. A 2 ms pulse represents a servo position of 180 degrees.

This pulse should be sent every 20 ms or so. The exact timing between pulses is not critical.

While we could easily write code to pulse the servo control the proper time (between 1 and 2 ms) every 20 ms, the Servo library used above takes care of that for us and can control multiple servos simultaneously.




Illustration 1: Source: seattlerobotics.org

NOTE: Servo range may vary; not all servos have a full 180 degree range.

Fail:

Unplug servo power line and plug it back in. Listen for servo to make a small move if power is connected properly.

Servos can consume more power than available from the Arduino and from laptop USB port. Try connecting the external battery pack for additional servo power.

Check that the values printed to the Serial Monitor make sense as the pot is moved.

Make sure the brown and red servo wires go to GND and +5v respectively (call instructor if unsure).

Make sure the orange wire is connected to the correct PWM pin on the Arduino, especially if there is more than one servo connected. Change to servo control pin defined in the software if necessary.

If the electrical connections are suspect, try replacing the 3 pin headers with 3 jumper wires

CONTROL: Robot DC Motor Control Using the KeyesL298 H-Bridge motor controller w/DC motors.

What:


Connect the left and right motors to the motor controller. Then to adjust the speed of each DC motor, you program the arduino to provide opposite HIGH/LOW direction signals into the motor controller's INput1/INput2 direction signals, then together with an EnableA (for motor-A for example) the A motor will begin spinning in one direction. Invert the IN1/IN2 signals to LOW/HIGH + ENA(HIGH), and it spins in the opposite direction. To control the motor speed, use a ~PWM output pin on the arduino and send it a PWM signal between 0-255 (0 off, 255 full speed) using `analogWrite(100)` (for example) to get a medium-slow spin.

How:

If you're using a two motor robot, (see photo) hook up motor-A to the L298's top motor terminals and the left motor to the motor-B terminals.

M.Board	Function	Sensor Shield
ENA	motor-a enable	D6
IN1	motor-a +	D7
IN2	motor-a -	D5
IN3	motor-b +	D4
IN4	motor-b -	D2
ENB	motor-b enable	D3
GND	Ground	GND
+5V	Logic Supply	VCC

Jumper	On/Off
CSB	ON
CSA	ON
UR1(pullup)	OFF
UR2(pullup)	OFF
UR3(pullup)	OFF
UR4(pullup)	OFF
5V_EN	OFF



The polarity of the motors (motor-a +, motor-a -) doesn't even really matter as the opposing motor will probably be turned around (180 degrees), so for your application you simply change the HIGH/LOW signals you send the IN1 and IN2 board inputs and that toggles the motor direction. Here's the example code for driving the motors:

```
#include <Arduino.h>

// Let's Code Blacksburg
// Motor Drive test #1 code
// (version .1 -ME )

// Motor controller signals and the arduino pin assignments
const int ENB=3;    // motor-b enable    PD3
const int IN4=2;    // motor-b -        PD2
const int IN3=4;    // motor-b +        PD4

const int IN2=5;    // motor-a -        PD5
const int IN1=7;    // motor-a +        PD7
```

```

const int ENA=6;    // motor-a enable PD6

void setup()
{
  // lchbb motor CONTROL test code.
  // Motor A setup
  pinMode (ENA,OUTPUT); //motor A enable
  pinMode (IN2,OUTPUT); //motor A wire 1 polarity
  pinMode (IN1,OUTPUT); //motor A wire 2 polarity

  // Motor B setup
  pinMode (ENB,OUTPUT); //motor B enable
  pinMode (IN4,OUTPUT); //motor B wire 1 polarity
  pinMode (IN3,OUTPUT); //motor B wire 2 polarity

  pinMode (13,OUTPUT); // LED for testing

  // Set polarity for motor A
  digitalWrite (IN4,LOW);
  digitalWrite (IN3,HIGH);

  // Set polarity for motor B
  digitalWrite (IN2,LOW);
  digitalWrite (IN1,HIGH);
}

void loop()
{
  int motorSpeed=255; // Any PWM range from 0-255

  digitalWrite(13,HIGH); // Turn on LED
  analogWrite(ENA,255); // set Motor A speed 100%
  analogWrite(ENB,255); // set Motor B speed 100%

  delay(3000);

  digitalWrite(13,LOW); // Turn off LED
  analogWrite(ENA,0); // set Motor A speed 0%
  analogWrite(ENB,0); // set Motor B speed 0%

  delay(3000);
}

```

If both of your motors are not turning on and off, you either have coding/typo problem or a hardware problem and need to see the **Fail:** section. If your motors are both turning on and off correctly, then try commenting out the last two **analogWrite** lines like this:

```

//analogWrite(ENA,0); // set Motor A speed 0%
//analogWrite(ENB,0); // set Motor B speed 0%

```

so your motors will be always on. Next flip your bot over, plug in the batteries to the arduino board and see if your bot goes in a straight line.

Q: Does your bot go in a solid straight line (with the code above)? A: _____

Not only can you vary the speed of each motor, but each motor has a unique, minimum usable speed (PWM) value where it begins to turn – under which it's really not usable. In fact, it's best to put the motors under slight load (touch them with your fingers or let it push the bot) and you'll see that minimum speed is even higher under load than at no load.

Flip your bot on it's back so you can watch the wheels and replace your **loop()** code block with the code below, and using the serial console (w/ the **Serial.print(motorspeed)** below), see what your minimum, loaded, motor speed is (the minimum PWM speed at which both wheels can move at roughly the same speed):

```
... (setup and other code) ...

void loop()
{
  int motorSpeed=0;           // starting speed
  int minSpeed=motorSpeed;    // minimum speed -\
  int maxSpeed=255;           // maximum, adjust -these to find what' best

  while (motorSpeed < maxSpeed) {
    motorRight(motorSpeed);
    motorLeft(motorSpeed);
    Serial.println(motorSpeed); // Print speed over serial
    motorSpeed+=2;             // increment the motor speed by 2
    delay (100);
  }

  while (motorSpeed > minSpeed ) {
    motorRight(motorSpeed);
    motorLeft(motorSpeed);
    motorSpeed-=2;             // decrement the motor speed by 2
    Serial.println(motorSpeed); // Print speed over serial
    delay (100);
  }
}

void motorRight(int speed)
{
  // Control the speed of Right motor-
  analogWrite(ENA,speed); // Send a PWM speed control to Right motorn
  // might have to swap with PD3 if your robot is wired differently than
  mine
}

void motorLeft(int speed)
{
  //Control the speed of motorB
  analogWrite(ENB,speed); // Send a PWM speed control to Left Motor
}
```

Looking at the serial console's output...

Q: What was your minimum usable motor speed (for both motors under load?) A: _____

Now set your **minSpeed** for that value, re-upload and see roughly where your bot travels in a straight line and make that value your new minSpeed and decrease your max speed to 2 higher than minSpeed and see if you can keep your bot running straight.

Q: What was your best straight line speed? A: _____

Fail:

One or both motors are not spinning at all.


- **Code:** Typo in the ENA, ENB or IN1/2/3/4 pin assignments
- **Hardware:** Double check your wiring (motor board to Arduino/Shield)
- **Hardware:** Wiring of the motors to the motor board, the power or the wires going to the motors
(have an instructor use a meter to check you motor resistance, motor wire/solder)
- **Hardware:** Your batteries are low (they should each measure between 3.7 – 4.2v)
(have an instructor use a meter to check your voltage levels)

INPUT: Ultrasonic “Eyes” Range Sensor

Using an ultrasonic ping sensor to detect object distance.

What:

The ultrasonic ping sensor is simply a digital module that allows you to send out a “ping” in the form of a microsecond sound burst, listen for the return echo and given the approximate speed of sound in air (using the delay between the ping and the echo back) calculate the distance to the object. Some ping sensors have a single ping/echo pin that you both send on and listen on (output + input), while others (such as the one we’re using here) have one pin dedicated to the ping (called the “trigger”) and one pin dedicated to receiving the echo (called “echo”). There is no analog signal processing that needs to be done as this is all handled by the module, so we’re just dealing with a nice clean, processed digital I/O signals and microsecond timing. However, the microsecond timing is very critical, and some cheaper sensors can not resolve distances much over a foot out.




How a ping sensor detects distance.

How:

Hook up the ping sensor to any two digital I/O pins free on your arduino. Here we’re using pins 12 (trigger) and 13 (echo). Be sure the Vcc (+5V) and GND are correctly wired before applying power.

No load up the build in example code from Examples / Sensors / Ping. This example code is made for the three pin Parallax version (with just one pin for trigger & echo), however you can easily modify it to work as seen below:



NOTE: If you have the US-100 version of this module, remove the jumper from the rear to make it use pulses instead of serial data.

```
#include <Arduino.h>

const int triggerPin = 12;
const int echoPin = 13;

void setup() {
  Serial.begin(9600);          // For outputting distance data
  pinMode(echoPin, INPUT);     // Make echoPin an INPUT
  pinMode(triggerPin, OUTPUT); // Make the triggerPin an OUTPUT
}

void loop()
{
  // establish variables for duration of the ping,
  // and the distance result in inches:
  long duration, inches;
```

```

// The PING))) is triggered by a HIGH pulse of 2 or more microseconds.
// Give a short LOW pulse beforehand to ensure a clean HIGH pulse:
pinMode(triggerPin, OUTPUT);
digitalWrite(triggerPin, LOW);
delayMicroseconds(2);
digitalWrite(triggerPin, HIGH);
delayMicroseconds(5);
digitalWrite(triggerPin, LOW);

// A HIGH
// pulse whose duration is the time (in microseconds) from the sending
// of the ping to the reception of its echo off of an object.
duration = pulseIn(echoPin, HIGH);


// convert the return echo time into a distance
inches = duration / 74 / 2;

Serial.print(inches);      // print it out to the serial port.
Serial.println("in, ");

delay(100);
}

```

Git code: https://github.com/LetsCodeBlacksburg/arduino-robotics/tree/master/LCBB_ping_trigger_echo_sensor_simple

After compiling and running this, then click on the serial monitor icon , you should get a stream of distances like this:

```

10in,
11in,
7in,
6in,
4in,
3in,
2in,
1in,
1in,
1in,
1in,
2in,

```

Think about at what distances you want to do things like (if doing a robot) slow your robot, stop your robot, or change directions.

Fail:

No distance readings can result from:

- bad wiring of power or trigger / echo pins
- not defining triggerPin as output and echoPin as input (or using them correctly)
- code typos

INPUT: Line Following IR Sensor

(Using a three-element infrared line sensor.)


What:

Line following robots need a way of detecting light vs dark lines on the floor. This three element digital sensor (with potentiometer threshold setting) is a popular low cost way of creating line following robots.

How:

Each line follower sensor has a Voltage pin (V), and Ground pin (G) and a Signal pin (S). One way to connect a three IR line follower circuits is by using the often unused analog inputs A0, A1 and A2 for the three signal lines from the sensors. The sensor's G & V pins can be connected to any ground and Vcc voltage sources. If using the Sensor Shield, these pins are usually free on any unused, three pin digital I/O connectors. The point here is to use the G & V pins on unused digital I/O connectors, but route the S (signal) lines over to an analog inputs, not the unused digital I/O S pin.

NOTE: The code below uses the serial monitor to watch all three modules. The serial monitor is a great way of watching the state of a lot of things at the same time like this.

Here's the example code for testing all three sensors while monitoring them over the serial monitor :

```
#include <Arduino.h>


// Let's Code Blacksburg
// IR sensor test code
// (version .2 -tweeks )
// This code reads the three IR line follower sensors
// (Left, Middle and Right) on analog pins A0, A1
// and A2, and outputs L for black line or _ for
// white surface (no line) on the serial monitor.

int sensorLeft=0;
int sensorMiddle=0;
int sensorRight=0;


void setup()
{
  Serial.begin(9600); // for outputting t-shooting codes
}

void loop()
{
  // Sensor readings give you a 1 if it sees the black line
  // and a 0 if it sees white.

  sensorLeft=digitalRead(A0); // Left uses analog A0
  sensorMiddle=digitalRead(A1); // Middle uses analog A1
```



Three of these make a



The wiring of the three element IR sensor

```

    sensorRight=digitalRead(A2);    // Right uses analog A2

    if (sensorLeft==1) {
        Serial.print(" L ");
    }
    else {
        Serial.print(" _ ");
    }

    if (sensorMiddle==1) {
        Serial.print(" L ");
    }
    else {
        Serial.print(" _ ");
    }

    if (sensorRight==1) {
        Serial.println(" L ");
    }
    else {
        Serial.println(" _ ");
    }
}

```

The latest version of this code can be found here:

https://github.com/LetsCodeBlacksburg/arduino-robotics/blob/master/LCBB_bot_tri_ir_line_sensor_simple/LCBB_bot_tri_ir_line_sensor_simple.ino

To test this, make a “testing card” out of a 3x5 card and black electrical tape. Place the bot on it's back and move the tape-card back and forth in front of the sensors.

Here's the example output of the serial monitor from this test code as you move the tape-card across the sensors, along with some comments I added indicating what you should tell your robot (motors) to do:

```

-   L   Go right
-   L L   Go right a little
-   L L   Go right a little
-   L L   Go right a little
-   L _   Go straight
-   L _   Go straight
-   L L _   Go left a little
L   _ _   Go left
L   _ _   Go left
-   _ _   Keep going the way you were before losing the line

```

Fail:

Common failure modes of these sensors are:

- **Hardware:** Hooking to the wrong pins (power or signal lines)
- **Hardware:** The underside of the board is shorting out on a metal surface (very bad)
- **Software:** Typos using analog reads instead of `digitalRead(A0)` (for example)

CONTROL: Potentiometer Control of a Servo's Position Recipe read a pot, move a servo.

What:

A potentiometer is just a variable resistor in the form of a knob. In order to use a potentiometer to control a servo:

- Follow the the potentiometer recipe, connecting the pot output to the Arduino pin A0 (analog 0).
- Follow the Servo recipe, connecting the servo input wire to Pin 9 of the Arduino.
- Use the program below

How:


Here is a sample of how it could be wired up.

After testing the Pot and servo separately use the following code to test Servo control by a potentiometer.

Turning the pot left and right should cause a corresponding rotation in the servo.

The servo may move “faster” or “slower” than the pot depending on the map in the code below.

(We are using the values 30 to 150 for the degree settings, because some servo's have difficulties at the extremes of their movement.)



```
#include <Servo.h> // servo library
// create a servo object
Servo servo0;

void setup() {
  // initialize serial communications at 9600 bps:
  Serial.begin(9600);
  servo0.attach(9); // servo is attached to pin 9
}

void loop() {
  int servo0Setting=90; // for the servo position later

  int sensorValue0;      // Read Pot value

  sensorValue0 = analogRead(A0);
  Serial.print ("Pot 0 value=");
  Serial.println(sensorValue0);
}
```



```

// Map the pot reading to the servo degree setting
// we'll use 30 to 150 degrees for the servo

servo0Setting=map(sensorValue0,0,1023,30,150);
Serial.print ("Servo 0 setting");
Serial.println (servo0Setting);

// Set Servo position
servo0.write(servo0Setting); // tell servo to go to the designated
position

}

```

Try This: You can change the direction of the servo to potentiometer mapping by swapping the pot input values in the map statement.

e.g. change this:

```

servo0Setting=map(sensorValue0, 0, 1023, 30, 150);
to this:
servo0Setting=map(sensorValue0, 1023, 0, 30, 150);

```

Of course different pins for the pot and servo may be used, but the test program will have to be modified to match.

Fail:

OUTPUT: Buzzer Alarms

Using passive piezo buzzers for beeps, buzzing and tones.

What:

Most of the time you're making sounds with an Arduino, all you need is beeps and buzzing noises. A passive piezo buzzer is perfect for this purpose. They can look like almost any of these devices on the right, but they're mostly the same in that, unlike speakers, they are high resistance and don't need an amplifier, high power or transistors to drive the device so they are very simple to implement.




One thing to watch out for is the use of active vs passive buzzers. Passive buzzers can be pulsed (with a signal from the Arduino) with almost any frequency (tone), very low or very high. Where active buzzers are “programmed” to only play one frequency/tone, usually around middle-C (or 1,000Hz). If you use a passive buzzer and send it a low tone (100Hz) you'll get a nice low tone sound. Send the same low tone to an active buzzer and there's no guarantee what you'll get, but it will probably sound like a mix between your tone and the preprogrammed tone.. kind of muddy and not clean sounding.

The gold device (right) is a raw piezo element and is guaranteed to be passive, where devices with the black casings (top) can be either active or passive and there's no way to tell which it is visually (if you didn't buy it). The down side to using gold/raw type of piezo element is that they cancel out some of their volume by not being encased in a nice sound directing casing (those black plastic cases).

The programs in this recipe are best used on passive, enclosed (black) piezo devices. But even still, if your tones sound dirty or garbled, chances are you have an active buzzer that does best with simple on/off signals instead of tone signals.

How:

Hooking up piezo buzzers is fairly simple. For a two wire buzzer (not shown here) you simply hook the buzzer's black wire to the Arduino's Gnd(G) and the red wire to one of the arduino's digital output pins. For the three-wire module variety (right), you simply hook the G/- pin goes to your Arduino's Gnd(G), the module's V/+ goes to Vcc/5v, and the Signal(S) pin goes to your preferred arduino digital output pin.



The code for making a single tone is as simple as:
(in setup)

```
pinMode(buzzerPin, OUTPUT);  
tone(buzzerPin, 3000, 500);
```

Which sends a single 3000Hz tone, on the buzzerPin (whatever you set that value to) for 500ms (or ½ a second).



No breadboard is needed if your arduino features SVG sensor headers!

Here's a little more advanced, and very useful recipe that you can use in many of your projects to generate human feedback on your projects:

```
const int buzzerPin = 3;           // Sound Buzzer on pin#

void setup() {
  pinMode(buzzerPin, OUTPUT);
  digitalWrite(buzzerPin, HIGH);   // Keeps buzzer from overheating and
                                   // making background noises.
}

///// LOOP /////
void loop() {
  buzzer("chirp"); delay(5000);
  buzzer("success"); delay(5000);
  buzzer("fail"); delay(5000);
  buzzer("alarm"); buzzer("alarm"); delay(5000);
}

///// buzzer("success" || "fail" || "chirp" || "alarm" ) /////
void buzzer(char mode[]){
  if(mode == "success"){
    tone(buzzerPin, 2000); delay(50); noTone(buzzerPin); delay(50);
    tone(buzzerPin, 2000); delay(50); noTone(buzzerPin);
  }

  if(mode == "fail"){
    tone(buzzerPin, 400, 400); delay(400);
    tone(buzzerPin, 100, 1500); delay(1500);
    noTone(buzzerPin);
  }

  if(mode == "chirp"){
    tone(buzzerPin, 3000); delay(50); noTone(buzzerPin);
  }

  if(mode == "alarm"){
    for (int x=0 ; x<1 ; x++){
      tone(buzzerPin, 1000, 1000); delay(1000);
      tone(buzzerPin, 800, 1000); delay(1000);
      noTone(buzzerPin);
    }
  }
}
```

Fail:


- The most common problem on the module device is hooking the pins up wrong.
- If a buzzer module sound sounds “muddy” or like two tones, you probably have an active (fixed frequency) version, which is not compatible with the `tone()` command.
- If the device gets hot to the touch, invert the `digitalWrite(buzzerPin, HIGH)` to LOW in setup, and use `noTone(buzzerPin)` after playing your tones.

OUTPUT: Sound Generation with Arduino

making tones, sounds and music with speaker


What:

Sounds are generated by feeding a changing voltage into a speaker. How often the voltage changes per second is the frequency of the sound. A “pure” sound would be a signal that changes smoothly over time – like a sine wave.



Sine waves can be hard to generate digitally. But for making basic tones, square waves are much easier to generate, require no real additional hardware and often works for most uses.

SQUARE WAVE



How:

Parts Needed:

- Speaker (The larger one in the kit, or a 5v buzzer.)
- Red and black wires.

Assembly:

- Connect the black wire from the black side of the speaker connector to the ground hole on the Arduino.
- Connect the red wire from the red side of the speaker connector to DIGITAL pin 5.

NOTE: In the LCBB arduino kit, the wires are a tight fit into the speaker connector. It may push the speaker wire out of the connector. So try to hold the speaker wire and connector while pushing in the other wires, or use the small, black 5v buzzer. If using a buzzer, be sure to check out the OUTPUT: Buzzer Alarms recipe.



The Code:

```
int soundOut = 5;      // Sound output pin
int del = 2;           // This determines how long the signal stays
                       // HIGH or LOW in milliseconds

void setup()
{
  pinMode(soundOut, OUTPUT);
}

void loop()
{
  digitalWrite(soundOut, HIGH);
  delay(del);
  digitalWrite(soundOut, LOW);
  delay(del);
}
```

}

Pretty simple, right? The delay is in milliseconds. So each pass through the loop is a complete “cycle” and bit more than 4 milliseconds total. There is some overhead to executing the various statements so that is why it comes out more than 4, but only by a little and it's close enough for us. That makes the frequency about 250 Hertz. That's close to a “B” an octave below middle “C” on a piano. Here's a chart for reference:

Note	1	2	3	4	5	6	7	8
B	62	123	247	494	988	1976	3951	
A sharp/B flat	58	117	233	466	932	1865	3729	
A	55	110	220	440	880	1760	3520	
G sharp/A flat	52	104	208	415	831	1661	3322	
G	49	98	196	392	784	1568	3136	
F sharp/G flat	46	92	185	370	740	1480	2960	
F	44	87	175	349	698	1397	2794	
E	41	82	165	330	659	1319	2637	
D sharp/E flat	39	78	156	311	622	1245	2489	4978
D	37	73	147	294	587	1175	2349	4698
C sharp/D flat	35	69	139	277	554	1109	2217	4434
C	33	65	131	262	523	1047	2093	4186

So let's modify it a bit to change tones back and forth – kind of like a European siren.

```

int soundOut = 5; // Sound output pin
int del = 1;      // Delay or duration of HIGH and LOW pulses
                  // Longer delay times = lower frequency
                  // So shorter delay 1 = high tone, longer delay 2 = low tone
int count = 0;    // Counter to control how long to play a high or low tone

void setup()
{
  pinMode(soundOut, OUTPUT);
}

void loop()
{
  digitalWrite(soundOut, HIGH);
  delay(del);

  digitalWrite(soundOut, LOW);
  delay(del);

  // Since del is the square wave's pulse time (inverse of frequency)
  // and count is a total amount of time, add del to count so we
  // always wait the same amount of time
  count += del;

  // If the desired amount of time as been reached, reset the
  // count and toggle the frequency (1 to 2)
  if(count > 500) // if past tone count (try 100, 25, 10, 4, 3, 2, 1 !)
  {
    count = 0;      // reset the duration counter
    if(del == 1)    // a == compares two things
    {
      del = 2;      // toggles the tone
    } else {
      del = 1;      // a single = sets a variable
    }
  }
}

```


Well, that was getting complicated! And if we want our Arduino to do anything else at the same time, it's going to get pretty nuts!

But remember the PWM info from the servos recipe? That's a square wave too! So lets try that.

```
int soundOut = 5;      // Sound output pin

void setup()
{
  pinMode(soundOut, OUTPUT);

  // Doing this here shows that we keep generating sound
  // Even after setup is complete and the loop can do
  // other things
  analogWrite(soundOut, 128);
}

void loop()
{
}
```

That's handy! See what happens if you change the analogWrite value.

But there's a limitation here. The PWM signal the Arduino generates is 490Hz. So you can't change the frequency.

Using **tone()** for Sound

So here's a better trick that greatly simplifies everything above...

```
int soundOut = 5;      // Sound output pin

void setup()
{
  pinMode(soundOut, OUTPUT);
}

void loop()
{
  tone(soundOut, 440);  // Surprise! Arduino can play sounds directly!
}
```

Wow, now that's handy! It's similar to the PWM technique in that the sound keeps playing by itself (try moving the “tone” command to the setup instead of loop). In fact, internally it uses the same hardware. But instead of a single frequency that varies the pulse width, you tell it the frequency and it's always a 50-50 high/low division. You can also add a third value for milliseconds (1000ms = 1 second) so it will only play for a given length of time. Pretty slick! Here's a tone chart.

Try This: Write twinkle twinkle little star! (try tone duration of 400ms and delay between notes of 500ms) and the chart above (starting in key of C).

So why even bother with other sound techniques? All technology has pros and cons. In this case, there are some limitations with the “tone” command. For one thing, it can only play one tone at a time. So even if you have multiple speakers hooked up, you can't send sound to all of them at once. Another problem is that since it uses some of the same hardware inside the Arduino, you can't use PWM on pins 3 or 11 at the same time you're generating tones. If these aren't a problem, and the sound quality is good enough, use tone.

If you're interested in these advanced techniques, check out “Advanced Arduino Sound Synthesis” in “Make:” magazine volume 35, page 80.


Fail:

OUTPUT: Playing MP3 Audio

Playing sounds/music w/the DFPlayer mini MP3 player module.

What:

Most of the time you're making sounds with an Arduino, all you need is beeps and buzzing noises. But if you need to play very high quality MP3 or .WAV audio files, then you need the DFPlayer, mini mp3 player with integrated audio amplifier. Although the DFPlayer module can function completely separately from an arduino (all music is stored on the removable microSD card), using the arduino with the mp3 module is nice in that it allows you to also do other things such as take button presses and send play/pause commands via the software serial driver, or even display track# information via an LCD or LED display.




How:

The library driver you need to install is the “DFRobotDFPlayerMini” + the SoftwareSerial library. Find and install these libraries through the Arduino IDE using its **Sketch / Include Library / Manage Libraries** interface. For example, just search for “DFPlayer” there and install the “DFRobotDFPlayerMini” driver, then the same for the SoftwareSerial library, and you’ll soon be able to program your arduino to send MP3 commands via the software serial lines.

NOTE: This recipe was written without a breadboard. Where you see the two 1k ohm resistors (below left of the speaker), you'll probably want to mount those on a breadboard and wire from the arduino to the resistors to the DFPlayer TX/RX serial lines.

!! WARNING !! Not using the serial data 1k resistors will likely blow your MP3 player module. Please don't let the magic smoke out of our our Arduino kits. ;)

You'll notice that this wiring diagram, or schematic, has a single sky blue wire running from Arduino pin 12 to the MP3 Player's active-low “_BUSY” line. This is so that we can (in code) monitor the high/low player-state of this line to know when the player is done playing (goes high), instead of bothering the MP3 player over it's serial line while playing music (which can cause some skipping and audio popping). We are also (optionally) taking one of the speaker outputs (light green line) and plugging it into the Arduino's analog input A1, which will let us monitor for audio (if we wish).



The code for this project is very long and so is not being fully provided in this cookbook at this time, but a variation of it can be copied and adjusted from a previous “Talking Skull” workshop that used an almost identical configuration. For that code, see the “LCBB_Talking-Skull_4_dfplayer-test” code located here in this github repo:

<https://github.com/LetsCodeBlacksburg/arduino-talking-skull>

INPUT: Light Sensor

detecting light intensity with a CdS photo-sensor

What:

A light sensing cell or photo-resistor or Cadmium Sulfide (CdS) cell turns light into a variable resistance. Usually dark = high resistance and high light = low resistance. In simple terms it is a bit like a potentiometer or volume knob for light.

Just like a variable pot resistor, it can be used as an input to change programming values for time delays, LED light intensity, or even motor movement or sound!

How:



Hooking up a raw light sensor is simple but does require an additional 10k ohm resistor wired in series (between +5V and GND on the breadboard). Your analog input (A5) gets its value from the voltage measurement from where the sensor and resistor meet on the breadboard. In this configuration (right), you can expect a data value of around 102 (of 1023, or roughly 0.5v) with a light shining on the sensor, around 716 (3.5v) for ambient light), and around 900 or more in complete dark.

Hooking up the light sensor *module* (instead of the raw sensor) is more simple. You simply provide it Vcc(V), Ground (G) and S (signal) into the same analog input (A5), but using the sensor headers if your arduino has them (right). Just be sure to note the your boards ordering of the S,V, and G pins.


Here's the code:

```
// With a 1k resistor in series with this light sensor, you should
// get around 900 range in the dark, around 600 in ambient, and
// around 2-200 range with an LED shining on it. Different values for
// different resistors. This is for a CdS photo resistor cell with
// approximate attributes of:
// DARK ~ 100k      AMBIENT ~ 2k      LIT ~ 0.7k

const int ls = A5;          //analog pin to which LS is connected
int ls_value = 0;           //variable to store LS values,
                           //we'll set this to zero first.
```



Wiring diagram for the raw light sensor and 10k resistor.
Source: http://wiki.xinchejian.com/wiki/Introduction_to_Arduino



No breadboard is needed if using a module sensor and your arduino has SVG headers!

```

void setup()
{
  Serial.begin(9600);  //Start the serial monitor
  pinMode(ls, INPUT);  //Tell arduino that this pin is for input
}

void loop()
{
  ls_value = analogRead(ls);  //reads the light sensor values
  Serial.println(ls_value);    //prints the LS values to serial monitor
  delay(50);                  //slow it down a bit
}

```

Try This: Ever hear of a Theremin? It's a musical instrument you can play by holding your hand over it! You can make one by combining this light detector with one of the previous sound or buzzer or sound recipes. Just hook up a buzzer or speaker to one of your digital output pins and add their `int` variable and setup sections to your own code and use the `tone()` line below in your loop:

```
tone(bunnerPin, ls_value, 10); //You just made a light Theremin
```

Tip: For a smoother sound, decrease your `delay(50);` line!

Note: If you're using a buzzer and getting strange (non-clean) garbled sounds, see the “OUTPUT: Buzzer Alarms” recipe on what can cause that.



Q: Do you like the sounds you can create with light?

Look over the recipe for “Ultrasonic Eyes” and see if you can make music with that sensor too!

Do you think it will sound any different?

Fail:

There's not a lot to go wrong with this setup, however common problems may include:


- Not having a raw sensor resistor and power hooked up correctly.
- Not tapping your A5 input reading off where the light sensor and resistor meet on the breadboard.
- Not having all your variables from the two programs both integrated, using the same variables names, a single unified `setup()` and `loop()` sections.

INPUT: DHT-22 Humidity/Temperature Sensor

Detecting relative humidity & temperature.

What:

The DHT-22 is a low cost, easy to use, and fairly accurate sensor for measuring both relative humidity (+- 5%) as well as temperature +0.5 °C) — all over a convenient “1 wire” serial interface. The inexpensive module can be run on either 3.3v or 5v, and the mounted “module version” (lower/left) includes built in pull up/conditioning resistors so that no breadboard or additional components are needed to simply connect it and get started!





How:

Connecting the DHT-22 module is accomplished via just three wires; two for power and one for the bi-directional, serial data or “1-wire” link (shown going into digital pin 4).

To connect (right), with the power off, simply connect the DHT module's GND to the Arduino's GND, Vcc to 5v, and the DATA pin to digital I/O pin 4 (or which ever digital I/O pin you want to use).

If your Arduino features SVG sensor headers, you can simply find the #4 SVG location and connect your:


- GND → G on the header
- +5V → V (middle, vertically), and
- DATA → S on the header.



No breadboard is needed if your arduino has SVG sensor headers!

However, before you can “talk” to the DHT, you need to first configure your Arduino environment with the proper library (created by Adafruit). To load the Adafruit DHT library, go to your Arduino IDE's **Sketch / Include Library / Manage Libraries** interface. One there, search for “DHT” and install the two latest libraries:

- 1) “DHT sensor library By Adafruit”
- +
2) “Adafruit Unified Sensor By Adafruit”



Once you have these in place, just load the example code (under the Arduino IDE's **File / Examples / DHT Sensor library / DHTtester**), or load this example code:

```
// Based on example sketch for various DHT humidity/temperature sensors
// by ladyada, public domain
#include "DHT.h"
#define DHTPIN 4           // what digital pin we're connected to
#define DHTTYPE DHT22      // DHT 22 (aka AM2302), AM2321

DHT dht(DHTPIN, DHTTYPE);
```

```

void setup() {
  Serial.begin(9600);
  Serial.println("DHT-22 test!");
  dht.begin();
}

void loop() {
  delay(2000);          // Wait a few seconds between measurements.
  // Reading temperature or humidity takes about 250 milliseconds!
  // Sensor readings may also be up to 2 seconds 'old' (its very slow)
  float h = dht.readHumidity();
  float t = dht.readTemperature();      // Read temperature as °C
  float f = dht.readTemperature(true);  // Read temperature as °F

  // Check if any reads failed and exit early (to try again).
  if (isnan(h) || isnan(t) || isnan(f)) {
    Serial.println("Failed to read from DHT sensor!");
    return;
  }

  float hif = dht.computeHeatIndex(f, h);    // Compute heat index °F
  float hic = dht.computeHeatIndex(t, h, false); // Compute heat index °C

  Serial.print("Humidity: "); Serial.print(h); Serial.print(" %\t");
  Serial.print("Temperature: "); Serial.print(t); Serial.print(" °C (");
  Serial.print(f); Serial.print(" °F)\t\t");
  Serial.print("Heat index: "); Serial.print(hic); Serial.print(" °C (");
  Serial.print(hif); Serial.println(" °F)");
}

```

Q: What happens when you run this?

Q: What happens when you breath on the sensor and when you stop?

Fail:

The most common problems with this circuit is getting your pins hooked up wrong. Always double check your +5v and Ground. Getting those wrong can “let the magi smoke out”.



- Check your wiring
- Make sure you program in a wait-time for your sensor to initialize
- Different readings than your neighbor? In yours in the sun? Under an AC vent?

INPUT: People Motion Sensor

PIR motion sensor for detecting motion up to 20ft away

What:


The “PIR” in PIR motion sensor stands for Passive Infra Red and motion sensor modules that passively detects body heat (or change in the temperature profile) in a visible space for up to 20ft away. Its operation is very simple and cheap to employ, and is commonly used in burglar alarms and outdoor motion activated lights.



How:

The PIR sensor is very simple to operate. It has only three pins. One power (+5V) pin, one ground (GND) pin and one active high OUT signal pin*. Once the module powers up and averages out its field of view (this takes between 15-30 seconds), then the OUT pin will go HIGH when there is motion and low a short time after the motion subsides.


*NOTE: This sensor runs on 5V, but sends out its OUTput signal at 3.3volts, which means that the most reliable setting for using it with an arduino is achieved by putting your arduino into 3.3v digital mode (usually a small 5v/3.3v switch on the arduino).



The L H mode jumper controls whether constant movement creates a constant HIGH condition (set to L) , or multiple HIGH/LOW pulses for as long as it senses motion (H mode). In most cases the L-mode is desired (HIGH for as long as there in movement). The delay-time and distance adjust (sensitivity) potentiometers should not be adjusted without speaking to an instructor first).

Connection is very strait forward, but we will connect it in this lab to the arduino's digital I/O pin #2.

The arduino requires no special software library to use this PIR module, but simply reads the OUT signal pin – HIGH for motion, LOW for no motion.. In fact, this sensor does not even need an arduino, but could be connected directly to a battery and power relay for controlling lights, space heaters, alarms, etc. However the arduino is convenient for giving such projects a little more intelligence and flexibility.



Here's the code for simply turning on and off a light with your PIR motion sensor:

```
////////////////////
//VARIABLES
int calibrationTime = 20; //the time we give the sensor to calibrate (20-30sec)
int pirPin = 2;          //the digital pin connected to the PIR sensor's output
int ledPin = 13;         // the built in LED pin

////////////////////
void setup(){
  Serial.begin(9600);
  pinMode(pirPin, INPUT);      // Set up the various I/O pins
  pinMode(ledPin, OUTPUT);
  digitalWrite(pirPin, LOW);

  //give the sensor some time to calibrate (monitor state on the serial port)
  Serial.print("calibrating sensor ");
  for(int i = 0; i < calibrationTime; i++){
    Serial.print(".");
    delay(1000);
  }
  Serial.println(" done");
  Serial.println("SENSOR ACTIVE");
  delay(50);
}

////////////////////
void loop(){

  if(digitalRead(pirPin) == HIGH){      // If the PIR is HIGH
    digitalWrite(ledPin, HIGH);         // then turn on the LED pin and
    Serial.println("HIGH (motion)");     // print "HIGH (motion)" on serial port
  }
  else {
    digitalWrite(ledPin, LOW);           // else, make the LED pin low and
    Serial.println("LOW (no motion)");   // print "LOW (no motion)" on serial port
  }
  delay(250);                           // 250mS delay
}
```

What else can you control or output using a motion sensor?

Fail:

The most common problems with this circuit is getting your pins hooked up wrong. BE SURE TO HAVE A TA CHECK YOUR WIRING BEFORE POWERING UP YOUR CIRCUIT!

- Check your wiring
- Make sure you program in a wait-time for your sensor to initialize
- Be sure your arduino is set to 3.3v mode and that you are sending 5V to power the sensor

OUTPUT: 7 Segment LED Display w/TM1637 Module Displaying “digital clock” numbers w/LEDs.

What:


The TM1637 is one of the quickest and easiest ways to employ a seven segment LED display in your projects. The units are small, cheap and very bright and usable in daylight. It has two power pins, and two I/O lines, a CLK which needs a clock input, and a DIO or digital I/O line that we communicate with the module with. It's very easy to use, but needs some special drivers that need to be manually loaded in your Arduino environment.




How:

The library/driver for this device in the Arduino library manager, can be problematic, so instead we recommend manually downloading and importing the **avishorp / TM1637** version of this driver, as it is a much better match for what we use these displays for at LCBB. To get this third party library installed, on the machine you're programming the Arduino with:

- 1) Browse out to the avishorp/TM1637 github page here: <https://github.com/avishorp/TM1637> and click the “Clone or Download” and the “Download ZIP” buttons.
- 2) Once the library download completes, in the Arduino interface, click **Sketch / Include Library / Add .ZIP** as seen to the right:



Once you get the driver installed, hookup is fairly stright forward. Like other digital modules used with the Arduino, you need to connect the module's VCC → 5V, GND → GND, and then the module's CLK (clock) and DIO (digital IO) to two unused Digital IO pins. Here in our schematic we're using pins 9 and 8 respectively (right).



Once you get it hooked up, load the TM1637Test fom the Arduino interface **File / Examples / TM1637 / TM1637Test**

After doing so, be sure to change the following two lines to match how you hooked up the hardware above:

```
#include <Arduino.h>
#include <TM1637Display.h>

// Module connection pins (Digital Pins)
#define CLK 8
#define DIO 9
...
```

(example code also available from <https://github.com/LetsCodeBlacksburg/arduino-recipes>)

INPUT / OUTPUT: The LCD / Keypad shield

Displaying text and taking keypress inputs

What:

By itself the Arduino has limited capability for interacting with humans. The LCD / Keypad shield provides a simple way to display information and allow direct user input through four direction and one select button.

NOTE: Shields are plug in arduino modules that are super easy to use, however their pins are easy to bend, so **please be careful and don't force the shield or be rough with it!**



How:

Hardware setup for shields is extremely easy – just power off/unplug the arduino, then carefully line up the shield pins and plug in the shield! The only gotcha is to be sure and align the pins on the shield correctly.

Lining up and Connecting/Disconnecting a Shield: There are fewer pins on the shield than there are holes in some Arduinos. To line up the pins correctly, hold the Arduino with the USB connection to your left and the shield with the buttons below the display in your right hand. Line up the farthest right shield pins with the farthest right Arduino holes. Carefully make sure all the remaining pins are aligned with the corresponding holes. Watch out for slightly bent pins! (call an instructor over if you see bent pins) Slowly squeeze the two boards together, being careful NOT to press on the display. To disconnect a shield, slowly rock it back and forth (end to end) while gently pulling it away from the arduino until it separates. Never just pry it off or you'll bend delicate pins!

Whenever you use any shield, it's important to understand what pins the shield actually use, both for being able to communicate with it and for knowing any conflicts which could occur if you want to stack shields. Below is a list of the pins used by the LCD / Keypad shield.

Pin	Function
Analog 0	Button (select, up, right, down, and left)
Digital 4	DB4
Digital 5	DB5
Digital 6	DB6
Digital 7	DB7
Digital 8	RS – Register Select (Data or Signal Display Selection)
Digital 9	Enable
Digital 10	Backlight Control

You might be tempted to stack the LCD shield on top of the Ethernet shield. However if you look at the pinouts you'll find that the Ethernet shield also uses pins 4 and 10. So you wouldn't be able to stack and use them at the same time. Even though they are located on the same board, the buttons and the LCD are logically separate and are handled separately. We'll start with the buttons.


Reading the buttons:

The buttons on the shield use a rather clever design. If you've connected buttons to the Arduino in the past, you probably connected them to digital inputs, maybe with a pull-up resistor. But that limits you to one button per pin. However this shield uses a single ANALOG pin for reading all the buttons. How is this possible?

The buttons are wired to a series of resistors so that when you press a button, a different voltage is presented to the analog 0 pin, and thus a different reading when you call `analogRead` for each button.

Enter the following code:

```
void setup() {  
  // initialize serial communication at 9600 bits per second:  
  Serial.begin(9600);  
}  
  
// the loop routine runs over and over again forever:  
void loop() {  
  // read the input on analog pin 0:  
  int buttonValue = analogRead(A0);  
  // print out the value you read:  
  Serial.println(buttonValue);  
  delay(1);          // delay in between reads for stability  
}
```

This code reads the analog pin and writes the result to the serial port. So after uploading this sketch, open the serial monitor in the IDE  and watch what happens.

Q: Press the buttons. What happens with the different buttons?

Did you press the right-most button? It may be labeled RST or RESET. This button is different from the others. This button is hard-wired as a RESET button. Pressing it will restart the Arduino, much like if you unplugged the power and then plugged it back in. You can't read or use it like the others.

Let's expand on the original button code a bit with a helpful function to decode those buttons into something a bit friendlier.

```
#define btnRIGHT  0
#define btnUP     1
#define btnDOWN   2
#define btnLEFT   3
#define btnSELECT 4
#define btnNONE   5

// read the buttons
int read_LCD_buttons()
{
  int adc_key_in = analogRead(0);      // read the value from the sensor
  // These values should work with all current keypads.
  // If you get unexpected results check the readings you get with you shield

  // We make this the 1st option for speed reasons
  // since it will be the most likely result
  if (adc_key_in > 1000) return btnNONE;
  if (adc_key_in < 50)   return btnRIGHT;
  if (adc_key_in < 250)  return btnUP;
  if (adc_key_in < 450)  return btnDOWN;
  if (adc_key_in < 650)  return btnLEFT;
  if (adc_key_in < 1000) return btnSELECT;
  return btnNONE; // when all others fail, return this...
}

// the setup routine runs once when you press reset:
void setup() {
  // initialize serial communication at 9600 bits per second:
  Serial.begin(9600);
}

// the loop routine runs over and over again forever:
void loop() {
  // read the input on analog pin 0:
  int sensorValue = read_LCD_buttons();
  // print out the value you read:
  Serial.println(sensorValue);
  delay(1);        // delay in between reads for stability
}
```

Enter, upload, and run the code as before. Now in your serial monitor window you should see '5' if nothing is pressed and other values between '0' and '4' when pressing a button.

Why not code in the exact values you saw in the first program? That might not be reliable or portable. While all the shields use the same common design and values, there is still some variation between parts. Each value should be within 20% of it's value, but unless something is REALLY wrong the values will be within a predictable range, so by mapping those values to a ranges we can make our program tolerant of differences between boards or temperature effects. So that's all there is to the buttons! So let's move to the more exciting bit – the LCD!

Writing to the LCD:

Now, you **could** attempt to program the LCD directly. It uses a standard HD44780 LCD controller in 4-bit mode if you want to look up how to do it. But unless you have some very special requirements you'll want to stick with the standard LiquidCrystal library that comes with the Arduino IDE. It makes using the LCD a breeze!

The full reference for the library can be found at <http://arduino.cc/en/Reference/LiquidCrystal>

The LiquidCrystal library is actually fairly generic. It will work with any HD44780 based display – you just have to tell it how it's connected. Using the shield the connection is fixed, but you still have to tell the library those connections – it can't tell automatically what you have plugged in.

Here's a simple program setting up and displaying a message and the time since the Arduino started up.

```
#include <LiquidCrystal.h>

// Create the LiquidCrystal object and define the pins used on the LCD
// shield
// The parameters are: rs, rw, enable, d4, d5, d6, d7
LiquidCrystal lcd(8, 9, 4, 5, 6, 7);

void setup()
{
  // Start the library, telling it we have a 16 column x 2 row display
  lcd.begin(16, 2);
  // Put the cursor on the first column of the first row
  lcd.setCursor(0, 0);
  lcd.print("Hello, world!");
  // First column of second row
  lcd.setCursor(0, 1);
  lcd.print("Seconds: ");
}

void loop()
{
  lcd.setCursor(9,1);    // move cursor to second line "1" and 9 spaces over
  lcd.print(millis()/1000); // display seconds elapsed since power-up
}
```

Pretty straightforward, right? There are a couple of “gotchas” to keep in mind:

1. Notice that, like most everything else in programming, the numbering of the rows and columns start with 0, not 1.
2. When you print something, it only prints EXACTLY what you tell it. If there was something else on the row before and your new text is shorter, it will only overwrite as much as the new text leaving behind a portion of the old. So you may need to pad out your strings with spaces to clear out any old output.

There are a lot more capabilities in the library – scrolling, blinking, and even creating special

characters. Look at the reference linked above for more information.

So for now, let's put all this together, reading buttons and displaying the result.

```
//Sample using LiquidCrystal library
#include <LiquidCrystal.h>

/*****
This program will test the LCD panel and the buttons
Mark Bramwell, July 2010
http://www.dfrobot.com/wiki/index.php?
title=Arduino_LCD_KeyPad_Shield_(SKU:_DFR0009)
*****/

// select the pins used on the LCD panel
LiquidCrystal lcd(8, 9, 4, 5, 6, 7);

// define some values used by the panel and buttons
int lcd_key    = 0;
int adc_key_in = 0;
#define btnRIGHT 0
#define btnUP    1
#define btnDOWN  2
#define btnLEFT  3
#define btnSELECT 4
#define btnNONE  5

// read the buttons
int read_LCD_buttons()
{
  adc_key_in = analogRead(0);      // read the value from the sensor
  if (adc_key_in > 1000) return btnNONE; // We make this the 1st option for
  speed reasons since it will be the most likely result
  if (adc_key_in < 50)   return btnRIGHT;
  if (adc_key_in < 250)  return btnUP;
  if (adc_key_in < 450)  return btnDOWN;
  if (adc_key_in < 650)  return btnLEFT;
  if (adc_key_in < 1000) return btnSELECT;

  return btnNONE; // when all others fail, return this...
}

void setup()
{
  lcd.begin(16, 2);
  lcd.setCursor(0,0);
  lcd.print("Push the buttons");
}

void loop()
{
  lcd.setCursor(9,1);          // move cursor to second line "1" and 9
```

```

spaces over
  lcd.print(millis()/1000);          // display seconds elapsed since power-up

  lcd.setCursor(0,1);               // move to the begining of the second line
  lcd_key = read_LCD_buttons();     // read the buttons

  switch (lcd_key)                  // depending on which button was pushed, we
  perform an action
  {
    case btnRIGHT:
    {
      lcd.print("RIGHT ");
      break;
    }
    case btnLEFT:
    {
      lcd.print("LEFT  ");
      break;
    }
    case btnUP:
    {
      lcd.print("UP    ");
      break;
    }
    case btnDOWN:
    {
      lcd.print("DOWN  ");
      break;
    }
    case btnSELECT:
    {
      lcd.print("SELECT");
      break;
    }
    case btnNONE:
    {
      lcd.print("NONE  ");
      break;
    }
  }
}
}

```

This code is from the DF Robot wiki page, combining reading the buttons with displaying some text in response.

Stretch Goal #1: Can you come up with a simple game using the buttons and the display? Perhaps start by moving a character around the screen in response to button presses!

Things to think about:

- When moving your character around, what has to happen in the character's old position?
- What will you do when you hit the end of a row?
- What will you do when you run off the top or bottom of the screen?

Fail:

COMM: Wired Web & Email Communications Using Ethernet networking for Web & Email

What:

The Ethernet shield gives your arduino projects a wired connection to a local network or the full Internet. The board (once connected) can either take a static or fetch a dynamic IP address (from a DHCP server or home router) and function as either a client (e.g. “browser”) or a server (e.g. web server, etc) — or even be used to send out emails, SMS/Text messages or even tweets!

The built in Wiz5100 networking chip does a very good job at offloading much of the complexities of running networking connections on the Arduino, but the required libraries do take up quite a bit of memory. So if you will be fine for doing very basic web, email or other basic networking communications, but anything more complex than that (dynamic web content, serving video or lots of images) and you'll want to start looking for a full computer platform with more horsepower, memory and networking support such as the RaspberryPi, Beaglebone or other embedded Linux Operating System.



How:

This module is wired to require pins 13, 12, 11, 10 for Ethernet usage, and pin 4 for SD card use (see Micro-SD card recipe).

This shield uses the Ethernet Library located here:




<http://arduino.cc/en/Reference/Ethernet>

Connecting and Disconnecting a Shield:

With your arduino unplugged, and on a static free surface, you will first want to carefully connect your Ethernet shield to your arduino, lining up the pins on the end opposite the shiny Ethernet and USB ports (see right, bottom). If your boards have the six ICSP pins, that is a good way of lining everything up (see photo on right) .

Once the shield is mounted, you can connect the Arduino/Laptop USB cable, and then the Ethernet cable. (photo at top of page).

To remove a shield, DO NOT just “pull it off”; you will bend our pins! Instead, gently half-way pry one side lose, then half way pry the other. Then gently lift off the shield and place it in a protective container.



Getting an IP Address:

The code below can be typed in, or loaded by using the example code Examples / Ethernet / DhcpAddressPrinter :

```
#include <SPI.h>
#include <Ethernet.h>

// Change at least the last hex number below to any two digit value so that
// you have a unique address on your class network. Otherwise, there will
// be MAC address conflicts and you and your neighbors will have problems.
byte mac[] = {
  0x00, 0xAA, 0xBB, 0xCC, 0xDE, 0x02 };

// Initialize the Ethernet server library
// with using DHCP for your IP (see serial monitor) and port you wish
// to use (port 80 is default for HTTP):
EthernetServer server(80);


void setup() {
  // Open serial communications and wait for port to open:
  Serial.begin(9600);
  while (!Serial) {
    ; // wait for serial port to connect. Needed for Leonardo only

    // start the Ethernet connection:
    if (Ethernet.begin(mac) == 0) {
      Serial.println("Failed to configure Ethernet using DHCP");
      // no point in carrying on, so do nothing forevermore:
      for(;;)
        ;
    }
  }

  // start the Ethernet connection and the server:
  Ethernet.begin(mac);
  server.begin();
  Serial.print("Arduino server is at ");
  Serial.println(Ethernet.localIP());
}

// Main execution loop
void loop() {
  Serial.print(".");    // Serial monitor heartbeat
  delay(1000);          // Simple 1 second do nothing
}
```

Quick Connectivity Test:

After compiling and uploading this code, click on the serial monitor icon , and copy or write down your IP in the serial console window (a four part number like 10.4.2.189).

If you are able to list your IP (above), then hook your laptop up to the same wired network, and try pinging your arduino like this:

```
ping -c 3 10.4.20.189          # Pings arduino three times (on Win use -n)
PING 10.4.20.171 (10.4.20.171) 56(84) bytes of data.
64 bytes from 10.4.20.171: icmp_seq=1 ttl=128 time=0.235 ms
64 bytes from 10.4.20.171: icmp_seq=2 ttl=128 time=0.252 ms
64 bytes from 10.4.20.171: icmp_seq=3 ttl=128 time=0.221 ms

--- 10.4.20.171 ping statistics ---
3 packets transmitted, 3 received, 0% packet loss, time 2000ms
rtt min/avg/max/mdev = 0.221/0.236/0.252/0.012 ms
```



Self Test Questions

Q: Did your ping test work?

Q: If so, what is your IP address? (write down on note pad)

If that worked, then your Arduino is officially on the network and you can now continue to to set it up as a little web server, email or twitter client. If problems, see the **Fail** section below.

Making a Mini Web Server:

Once you have the ping/connectivity test above working, you're ready to start serving web content or real time input sensor data. First, we need to replace the entire main loop to 1) detect client connections, 2) write out the HTTP web headers, 3) print some basic web content (test text). See below:

```
// Main execution loop
void loop() {
  // listen for incoming clients
  EthernetClient client = server.available();
  // If a client has connected, do this
  if (client) {
    Serial.println("new client");
    // an http request ends with a blank line
    boolean currentLineIsBlank = true;
    while (client.connected()) {
      if (client.available()) {
        char c = client.read();
        Serial.write(c);
        // if you've gotten to the end of the line (received a newline
        // character) and the line is blank, the http request has ended,
        // so you can send a reply
        if (c == '\n' && currentLineIsBlank) {
          // send a standard http response header
          client.println("HTTP/1.1 200 OK");
          client.println("Content-Type: text/html");
          client.println("Connection: close"); // the connection will be closed after gets response
          client.println("Refresh: 5"); // refresh the page automatically every 5 sec
          client.println();
          client.println("<!DOCTYPE HTML>");
          client.println("<html>"); // Opens HTML content
          client.println("<pre>"); // Preformatted text block
          client.println("TEST... SUCCESS!!!"); // Actually printed to browser screen
          client.println("</pre>");
          client.println("</html>"); // closes out html web page
          break;
        }
        if (c == '\n') {
          // you're starting a new line
          currentLineIsBlank = true;
        }
        else if (c != '\r') {
          // you've gotten a character on the current line
          currentLineIsBlank = false;
        }
      }
    }
    // give the web browser time to receive the data
    delay(10);
    // close the connection:
    client.stop();
    Serial.println("client disconnected");
  }
}
```

Congratulations! If you have **TEST... SUCCESS!!!** on your browser screen, then your Arduino is now serving static web content.

Q: Did you get your arduino to display the test web page?

Displaying Analog Readings:

After you get the basic web content bring served from your arduino, next try serving out some analog voltages values from the six analog inputs 0 – 5. Find the “**TEST... SUCCESS!!!**” line in the **HTML content code area** of your current code and replace the whole line with this code below

```
client.println("=====");
client.println("=== Tweeks Arduino Webserver ===");
client.println("=====");
// output the value of each analog input pin
for (int analogChannel = 0; analogChannel < 6; analogChannel++) {
  int sensorReading = analogRead(analogChannel);
  client.print("analog input ");
  client.print(analogChannel);
  client.print(" is ");
  // Adjust the for loop, adjust "n" of x<sensorReading/n to
  // scale the 1023 bar graph (play with it)
  int x=0;
  for (x=0 ; x<sensorReading/15 ; x=x+1) {
    client.print("=");    // builds your ASCII TXT bar graph
  }
  client.print(sensorReading); // prints the actual value
  client.print("<br />");
}
client.println("</pre>");
client.println("</html>");    // closes out html web page
```

If you got it working, your web browser should be displaying real time graphs of your six analog input readings like this:

```
=====
=== Tweeks Arduino Webserver ===
=====
analog input 0 is =====310
analog input 1 is =====328
analog input 2 is =====313
analog input 3 is =====346
analog input 4 is =====301
analog input 5 is =====314
```

Display Real World Values on The Web:

Want to read a real value? Look at the “**What**” sections of the **INPUT: Potentiometer Recipe** to hook up a potentiometer (knob) to Analog input A0. Look at the “What” and “How” sections of **INPUT: Light Sensor** to measure the amount of ambient light in the room.

Stretch Goal #1: Can you make your pot (knob) value print out to be “loud” “medium” or “soft”, or your light sensor value print out to be “lights on” or “lights off”?

Stretch Goal #2: Want cooler looking graphs and text windows? Check out the HTML codes you can use in place of the “=” signs here (hint, use the `&code;` format to `client.print()` :

<http://www.theasciicode.com.ar/extended-ascii-code/bottom-half-block-ascii-code-220.html>

Sending Email:

Email is still being tested, but here's the code you can start playing with if email is your desire:

To use it for the Nov 20th 2014 workshop, see the guest instructor Steve Swenson for a guest SMTP username, password for relaying (or sending through) email out of his SMTP accounts. For this November workshop, also:

- Change line 13 to smtp.emailsrvr.com, keep the port at 587.
- Change line 73 to add your base64 encoded email address, so it should look like:
 "AUTH login -insert-your-base64-username-here-"
- Change line 76 to contain your base64 encoded password
- Change line 79 and 89 to the from address (-username-@rackeddit.net)
- Change line 82 and 90 to the TO address, e.g. me@gmail.com
- Everyone must chose a unique MAC address on line 107
- Suspect the function on line 38 isn't used. You can cut it to save some bytes.
- If you need to base64 encode anything else, you can use this fiddle:
 <http://jsfiddle.net/pb4usf47/>

Here's the code:

(for line numbers, reference the github code here:

<https://gist.github.com/ctigeek/a6ac72253ddd588e9b35#file-emailtest-ino>

```
// Arduino script for sending an email by @ctigeek.

#include <SPI.h>
#include <Ethernet.h>

EthernetClient  client;
signed long next;

int connectToServer() {
    Serial.println(F("Connecting to server..."));

    if (!client.connect("smtp.server.com",587))
        return 0;

    Serial.println(F("connected to mail server."));
    return 1;
}

int waitForReply() {
    next = millis() + 5000;
    while(client.available()==0)
    {
        if (next - millis() < 0)
            return 0;
    }
    int size;
    while((size = client.available()) > 0)
    {
        uint8_t* msg = (uint8_t*)malloc(size);
```

```

        size = client.read(msg,size);
        Serial.write(msg,size);
        free(msg);
    }
    return 1;
}

int sendAndWaitForReply(char* txtToSend) {
    next = millis() + 5000;
    client.println(txtToSend);
    Serial.println(txtToSend);
    if (!waitForReply())
        return 0;

    return 1;
}

void justSend(const __FlashStringHelper* txtToSend) {
    client.println(txtToSend);
    Serial.println(txtToSend);
}

int sendAndWaitForReply(const __FlashStringHelper* txtToSend) {
    next = millis() + 5000;
    client.println(txtToSend);
    Serial.println(txtToSend);
    if (!waitForReply())
        return 0;

    return 1;
}

int sendEmail() {
    if (!connectToServer())
        return 0;

    if (!waitForReply())
        return 0;

    if (!sendAndWaitForReply(F("EHLO Arduino"))))
        return 0;

    if (!sendAndWaitForReply(F("AUTH login "))) //enter base64 encoded email
address here
        return 0;

    if (!sendAndWaitForReply(F(""))) //enter base64 encoded password here
        return 0;

    if (!sendAndWaitForReply(F("MAIL FROM:<your email address here>"))))
        return 0;

    if (!sendAndWaitForReply(F("RCPT TO:<recipients email address here.>"))))
        return 0;
}

```

```

    if (!sendAndWaitForReply(F("DATA")))
        return 0;

    justSend(F("MIME-Version: 1.0"));
    justSend(F("From: sender@domain.com"));
    justSend(F("To: recipient@domain.com"));
    justSend(F("Subject: Arduino!"));
    justSend(F("Content-type: text/plain; charset=us-ascii"));
    justSend(F("")); //this separates the headers from the body.

    justSend(F("Arduinos are so small!"));

    if (!sendAndWaitForReply(F("."))) //this means EOF to the smtp server.
        return 0;
    if (!sendAndWaitForReply(F("QUIT")))
        return 0;
    return 1;
}

void setup() {
    Serial.begin(9600);
    uint8_t mac[6] = {0x00,0x01,0x02,0x03,0x04,0x05};
    Ethernet.begin(mac);
    Serial.print(F("localIP: "));
    Serial.println(Ethernet.localIP());
    sendEmail();
}

void loop() {
}

```

Fail:

For the Web portions of this recipe:

- If Failed the “Quick Connectivity Test”, then:
 - Check that ethernet cable is connected and you have a link light on the arduino and switch
 - Check that the lights on your ethernet board are properly lit (shield is properly hooked up).
 - Did you GET an IP address printed to the serial console?
 - Is your laptop hooked to the same wired network?
 - Check for error messages during the build (indicating typos). CODE IS CAPS SENSITIVE!
 - Look at how your neighbor got it working or ask the TA for help
- If Failed the “Making a Mini Web Server” “TEST.. SUCCCESS” (but passed the ping test):
 - PEBKAC... Check for typos.
- If Failed the “Displaying Analog Readings” section, then:
 - PEBKAC... Check for typos
- If you can't find typos, but it isn't working, copy/paste the example code from github, or the student thumb drive (if the instructor has code locally).


COMM: Wireless Web & Email Communications Using WiFi ESP-01 for Web & Email

What:

The ESP-01 (or ESP8266 ver 01) is a very low cost (around \$1!), stand alone microcontroller that handles WiFi communication and networking all by itself, and can be “connected to” from other microcontrollers that can only speak serial send(TX) / receive(RX) data. This makes it a simple and cheap way of getting an Arduino (or other embedded system) onto a local wireless network, the Internet or even the web!



What we're going to do here is simply get this ESP-01 module (shown here with a blue 5v-to-3.3v adapter board) onto the local wireless network, take one or two simple analog readings, and “publish” them to a mini web server we will run from the Arduino, out to the ESP-01 (via software serial link) to be served on www/port 80 to the Internet! Pretty cool, right?!



*NOTE: To get any device onto the local wireless LAN, you need to have access. Unless it is a completely open/public (no login) network, this means either a valid username/password, or your device's MAC address must already be registered be “be allowed” to get on line. This workshop has already taken measures to ensure you can get on. See your instructor for any special SSID/login info. Also know, the drivers used in this recipe only work with WAP, WPA, and some WPA2 (SSID + network password). It does **not** work with WPA2-Enterprise (unique username/password logins).*

How:

The raw ESP-01 device does not use a standard pitch connector and it does not play well with breadboards, and so physically connecting one can be a challenge. Add to that that it is also a 3.3V device and can be easily blown if connected directly to a 5v Arduino.



TIP: The ESP-01 and other ESP modules like to communicate at a default serial “baud” rate of 115200. The ArduinoMega can handle that speed, but the Uno and smaller Arduinos can not. Before you can use any off the shelf ESP-01 with an Uno (or similar), you need to configure the ESP unit's UART BAUD rate to 9600, and do all serial data (to the arduino and to the ESP module using SoftwareSerial) at 9600 baud. This is a common mistake if doing this at home. If you not not using a LCBB supplied ESP-01, you need to set this with the “AT + UART_DEF = 9600,8,1,0,0” command to the ESP-01. Google for Uno, ESP-01, 9600, UART_DEF more information.

Lucky for us, Let's Code's ESP-01's have all been set to 9600 baud, and we're also using these convenient adapter boards (small blue board under the ESP-01) that makes it super easy to connect, **plus** has a 5-to-3.3V regulator, **and** 3.3/5v voltage converters for the serial data lines!

To connect our ESP-01 module using its special carrier adapter board only takes four wires.. two for power and two for serial send (TX) and receive (RX). If your Arduino has an SVG sensor/servo header (see right), then it makes things even easier as no breadboard (for power) is needed.

Here's how to hook it up:

- ESP GND → to GND on Arduino (or G on the SVG header)
- ESP VCC → 5V on arduino (or V on the SVG header)
- ESP TX → to pin 6 (or pin 6's S on the SVG header)
- ESP RX → to pin 7 (or pin 7's S on the SVG header)



No breadboard is needed if your arduino has SVG sensor headers!

To be able to serve some sort of “data”, it's also useful to hook up a simple light sensor or potentiometer to an analog input so that you can monitor some real, physical thing from the mini-web server we're going to run, but this is not required. In this example we just monitor the random analog input data on analog pin A5 and call it “light”, but you can use the “Light Sensor” recipe and measure it for real.


Special (not-stock) WiFiESPLibrary Install:

We recommend against installing the WiFiEsp (by bportaluri) from the Arduino's built in library search/add tool. It has some dependency issues that will cause compile issues on various platforms. Instead, install a fixed up version of the original WiFiEsp library available from Tweeks' github fork of this ESP library. To get the correct/working version, download this .ZIP archive of the library, browse out to Tweeks' github repo to his fork of WiFiEsp here: <https://github.com/Tweeks-va/WiFiEsp>

Next, click on Clone or Download / Download ZIP icon.

Once the download is complete, in the Arduino IDE import the downloaded library zip file with the Arduino IDE's “**Add .ZIP Library**” feature (shown right).

Now let's start getting our Arduino on line...



Below is the code to implement our miniature WiFi web server enabled Arduino. If you have problems getting it all poked in accurately, the barebones functional version can be downloaded from the LCBB arduino-recipes repository at <https://github.com/LetsCodeBlacksburg/arduino-recipes> (click on wifi-web-esp-01-sensors) and copy the raw code into a new Arduino IDE window:

NOTE: You MUST already be able to get on your local wifi network and provide the SSID (network name) and password in the code below.

```
// Wireless Libraries & Variables
////////////////////////////////////
#include <WiFiEsp.h>
#include <WiFiEspClient.h>
#include <WiFiEspServer.h>
#include <WiFiEspUdp.h>
#include "WiFiEsp.h"
// Emulate Serial1 on pins 6/7 if not present
#ifndef HAVE_HWSERIAL1
#include "SoftwareSerial.h"
SoftwareSerial Serial1(6, 7); // RX, TX
#endif
char ssid[] = "LOCAL-WIFI-SSID";          // the network SSID (name) you're
connecting to
char pass[] = "LOCAL-WIFI-PASSWD";        // the network password (blank if
none)
int status = WL_IDLE_STATUS;              // the Wifi radio's status
int reqCount = 0;                          // number of requests received
WiFiEspServer server(80);

const int buzzerPin = 3;                   // Sound Buzzer

// See Light Sensor recipe
char sensorLightHeader[] = "Current Light Level (0-1023) = ";
int sensorLightData = 0;

// See DHT-22 recipe to make work
char sensorTempHeader[] = "Current Temperature(°C) = ";
float sensorTempData = 0;

// See DHT-22 recipe
char sensorHumidHeader[] = "Current Humidity(%) = ";
float sensorHumidData = 0;

////////// Setup
////////////////////////////////////
void setup()
{
  pinMode(buzzerPin, OUTPUT);              // Sets buzzerPin as output
  digitalWrite(buzzerPin, HIGH);

  ////////// Setup WiFi
  Serial.begin(9600);                      // initialize serial for debugging
```

```

Serial1.begin(9600);      // initialize serial for ESP module
WiFi.init(&Serial1);      // initialize ESP module
if (WiFi.status() == WL_NO_SHIELD) {    // check for ESP-01
    Serial.println("WiFi shield not present");
    //buzzer("fail");      // hmmm.. this would be useful if it worked.
    while (true);          // don't continue
}
// attempt to connect to WiFi network
while ( status != WL_CONNECTED) {
    Serial.print("Attempting to connect to WPA SSID: ");
    Serial.println(ssid);
    status = WiFi.begin(ssid, pass);    // Connect to WPA/WPA2 network
}
tone(buzzerPin, 2000, 50); delay(50); noTone(buzzerPin); // Connected
Serial.println("You're connected to the network");
printWifiStatus();
server.begin();           // start the web server on port 80
}

//// loop()
////////////////////////////////////
void loop()
{
serveWebPage();    // This is all we do here. :)
}

//// serveWebPage
////////////////////////////////////
void serveWebPage(){
    // listen for incoming clients
    WiFiEspClient client = server.available();
    if (client) {
        Serial.println("New client");
        // an http request ends with a blank line
        boolean currentLineIsBlank = true;
        while (client.connected()) {
            if (client.available()) {
                char c = client.read();
                Serial.write(c);
                // if you've gotten to the end of the line (received a newline
                // character) and the line is blank, the http request has ended,
                // so you can send a reply
                if (c == '\n' && currentLineIsBlank) {
                    sampleSensorData();
                    Serial.println("Sending response");

                    // send a standard http response header
                    // use \r\n instead of many println statements for speed
                    client.print(
                        "HTTP/1.1 200 OK\r\n"
                        "Content-Type: text/html\r\n"
                        "Connection: close\r\n" // connection will be closed > response

```

```

        "Refresh: 20\r\n"          // auto-page refresh every 20 sec
        "\r\n");
    client.print("<!DOCTYPE HTML>\r\n");
    client.print("<html>\r\n");
    client.print("<h1>My WiFi Arduino!</h1>\r\n");
    client.print("Requests received: ");
    client.print(++reqCount);
    client.print("<br>\r\n");
    client.print(sensorLightHeader);
    client.print(sensorLightData);
    client.print("<br>\r\n");
    // put more sensors here
    client.print("</html>\r\n");
    //buzzer("chirp");          // See buzzer Alarms recipe if interested
    break;
}
if (c == '\n') {
    // you're starting a new line
    currentLineIsBlank = true;
}
else if (c != '\r') {
    // you've gotten a character on the current line
    currentLineIsBlank = false;
}
}
}
// give the web browser time to receive the data
delay(100);

// close the connection:
client.stop();
Serial.println("Client disconnected");
}
}

///// printWiFiStatus
////////////////////////////////////
void printWiFiStatus()
{
    // print the SSID of the network you're attached to
    Serial.print("SSID: ");
    Serial.println(WiFi.SSID());

    // print your WiFi shield's IP address
    IPAddress ip = WiFi.localIP();
    Serial.print("IP Address: ");
    Serial.println(ip);

    // print where to go in the browser
    Serial.println();
    Serial.print("To see this page in action, open a browser to http://");
    Serial.println(ip);
}

```

```

    Serial.println();
    // make a chirp sound once on network
    tone(buzzerPin, 2000, 50); delay(50); noTone(buzzerPin);
}

///// sampleSensorData()
////////////////////////////////////
void sampleSensorData(){
    // This is where you do your light, temperature and humidity sampling
    (DHT-22 could take TWO SECONDS)

    // Light Sensor (see receipe for more info)
    sensorLightData = analogRead(A5);

    // DHT-22 Humidity/Temp Sensor (see recipe for more info)

    // Other sensors to sample?
    tone(buzzerPin, 1000, 50); delay(50); noTone(buzzerPin);
}

```

Stretch Goal #1: Can you add a real light sensor, as well as other sensor types? What types of physical things would you like to monitor in real time? Temperature? Humidity? Combine this with the DHT-22 Humidity/Temperature recipe to do just that!

Stretch Goal #2: Instead of just reporting TXT sensor values, how about doing data-graphs? Check out the for loops used to build data value graphs in the “Wired Web” recipe before this one. How cool can you make your web sensor data look?

Stretch Goal #3: Email was left out of this recipe, but Tweeks' library includes an SMTPClient.h function for talking to SMTP (email) servers. Can you figure how to get it to work? If so, use this to send an email, or an email to an SMS (TXT message) gateway (e.g. phone#@mywirelessprovider.com usually works).

Fail:

Common failures for this recipe are:

- Arduino serial & SoftwareSerial baud rates are not BOTH set to 9600 (must be)
- Bad SSID / password plugged into code (typos)
- Closed network not allowing your SSID to join
- WP2-Enterprise w/individual username/passwords (instead of a shared SSID password)
- Network requires MAC address registration
- Poor wifi coverage (the antenna in the ESP-01 isn't that great. Check out the WiFiEsp examples of “BasicTest” or ScanNetworks to see all AP RF levels)

Recipe Wish List

Future Recipes & Modules to Add

Arduino Coding/Theory
 programming sections
 reference commands
 coding best practices
IR Remote Control (w/dedicated thin remote)
Bluetooth Remote Control (w/phones)

Future Eletronics Lessons:

Using a multi-meter
Light and external LEDs
 20ma limits
 resistor
 pin modes
 dim light means port wasn't set to output
 pwm
Drive a motor / transistor
 20ma limits
 biasing a transistor
 pin modes
 pwm
Resistors and Ohms law recipe
soldering skills

far future

GPS
9 DOF GYRO
USB hids devices